
C H A P T E R

THE SPECIFICS oN SUBFILES

ccording to IBM, a subfi le is a group of records that have the same record 
format and are read from and written to a display station in one operation. As 
this defi nition suggests, a subfi le is not a fi le; rather, it is a temporary place to 

store data of the same format to be used by a display fi le program.

Why Use Subfi les?

Figure 1.1 shows the typical use of a subfi le. As you can see, subfi les are useful when 
you want to list multiple, like records on a display. A major benefi t of subfi les is that 
you can defi ne them so the number of displayed records exceeds the number of lines 
available on the screen, allowing the user to scroll, or page, through the data. This is 
usually the reason you would decide to use subfi les in your display program. Because 
of their ease of use, however, I deploy subfi les even when I think I will never display 
more than one screen, because they are easy to change if the data someday requires 
more than one screen.

Figure 1.1: A typical subfi le application lists records from a fi le and allows the user to page 
through them.

In addition to simply displaying multiple lines of data, you can use subfi les and their 
multiple-line capabilities to add, change, and delete records very effectively. You can 
also use subfi les in a non-display manner, to create self-extending data structures and 
arrays in your RPG program. They work like multiple-occurrence data structures and 
arrays, except that instead of having to hard-code a number of elements large enough 
to hold the maximum number of entries, you can use subfi les. You start with a small 
number of entries and allow the number to expand dynamically, as your data expands.

Why Should You Care?

Haven’t programmers been able to display data to a screen for years, without subfi les? 
Yes, but subfi les allow you to display lists of like data that can extend beyond one 
screen. They make it easier for you to create display applications, which, in turn, 
makes you more productive. Subfi le programs are easy to write and maintain because 
much of the work is done for you in the Data Defi nition Specifi cations (DDS). Most 

of the time, you can change the characteristics of a subfi le program without having to 
modify the RPG code driving it.

If that isn’t enough, keep in mind that the ability to code subfi les is often the measure 
of an RPG programmer’s worth. As an added benefi t, when you’re talking with a 
group of programmers at a users’ group or technical conference, you can hold your 
head high, knowing that you’re right up there with the rest of the subfi le-savvy 
programmers.

Two DDS Formats Are Better Than One

As I stated earlier, most of the work in a subfi le application is accomplished in the 
DDS. For every subfi le you describe in your DDS, you’re required to use two format 
types: a subfi le record format (SFL) and a subfi le control record format (SFLCTL). The 
subfi le record format is used much as a record format is for a physical fi le. It defi nes 
the fi elds contained in a row of a subfi le, describes the defi nition and attributes of those 
fi elds, and holds the actual data.

Unlike a physical fi le, however, a subfi le record format is used only in memory, and 
only for the duration of the job. Once the program using the subfi le ends, the data in 
that subfi le is gone. Individual subfi le records are read from, written to, and updated to 
the subfi le by accessing the subfi le record format in your program. The SFL keyword 
is required in your DDS to defi ne a subfi le record, just as the RECORD keyword is used 
for a typical display record format.

The subfi le control record format describes the heading information for the subfi le and 
controls how the subfi le is going to perform. It’s here that you defi ne the size of the 
subfi le, as well as the conditions under which the subfi le will be cleared, initialized, 
displayed, and ended. The subfi le control record format is unique because it controls 
aspects of a subfi le in a way that other fi les aren’t controlled. For example, you control 
the size of a physical fi le during compilation or by using the Change Physical File 
(CHGPF) command. You do not determine it in the DDS. Your program will operate 
the subfi le control record format directly when it performs functions to the whole 
subfi le, not individual records. Actions such as initializing, clearing, and displaying are 
accomplished in this fashion.

Four keywords are required in the subfi le control record format:

1. The subfi le control (SFLCTL) keyword identifi es the subfi le control record 
format, much as the record (RECORD) keyword identifi es a typical display record 
format. The SFLCTL keyword also identifi es the subfi le record format that must 
immediately precede it.

2. The subfi le size (SFLSIZ) keyword specifi es the initial number of records the 
subfi le may contain.

3. The subfi le page (SFLPAG) keyword specifi es the number of records that one 
screen of data may contain.

4. The subfi le display (SFLDSP) keyword specifi es when the subfi le is displayed.

Note that the fi rst requirement states that the SFLCTL keyword identifi es the subfi le 
record format that precedes it. Therefore, the SFL record format must immediately 
precede the SFLCTL format. The individual keywords available in each format can be 
placed in any order, but they must exist at the top of the format, before any constant or 
fi eld defi nitions.

DDS and RPG Working as One

Once you code and compile the DDS for a subfi le, you are ready to use that subfi le in 
your RPG program. Your fi rst three actions will be initializing, loading, and displaying 
your subfi le. There are other things you can, and will, do to a subfi le, but let’s start 
with that foundation, and build up from there.

Before loading a subfi le, you might want to clear or initialize it for use. Clearing 
or initializing takes effect when you write to the subfi le control record format with 
the appropriate conditioning indicator set on, using the RPG WRITE operation. This 
isn’t a requirement for the introductory load in your program, but as you create more 
complex subfi le applications, you’ll need to know the difference between clearing and 
initializing a subfi le.

Let’s return to the DDS for a moment. To clear a subfi le, use the subfi le clear (SFLCLR) 
keyword. This removes all records from the subfi le. However, it does not remove 

them from the display until the next time the subfi le is written to the screen by your 
program. This is different from the subfi le initialize (SFLINZ) keyword, which will 
set all the records in your subfi le to their default value. If you have no default value 
set in your DDS, numeric fi elds will default to zeros and character fi elds will default 
to blanks. Suppose your subfi le size keyword is set to 100. If you clear it, instead of 
having an empty subfi le, you will have 100 records, each initialized with its default 
value. I will talk more about SFLINZ later on, so don’t worry if you don’t quite get it 
yet. In short, the decision about whether to use SFLCLR or SFLINZ is determined by the 
role of your program.

You can load your subfi le in a few different ways. The most common way is to retrieve 
data from one or more data fi les and write that data to the subfi le record format, using 
the RPG WRITE operation. You might also choose to write only enough records to fi t 
on one page (as determined by SFLPAG), and write more only if the user requests more; 
write all the desired records to the subfi le before displaying the screen to the user; or 
use a combination of both. Still another way to load your subfi le is to initialize the 
subfi le fi rst using the SFLINZ keyword, display the subfi le with its default values to the 
screen, and allow the user to type data into the subfi le from the screen.

As we go along in this book, I will show you examples of all these techniques because 
they are coded quite differently. I will also explain when to use which method.

Once your subfi le is loaded or initialized, you’re ready to display it. Depending on 
how you created your DDS, your subfi le can take on many looks. No matter which 
form it takes, however, you only need one line of RPG code to display it.

Let’s look at the code for a basic subfi le program that displays a list of names. These 
could be the names of customers, salespeople, or employees; for our purpose, we’ll 
just use generic names. Take notice of how much work is done with very little RPG 
code. Be careful, though. Once you understand this code, there’s no turning back—
you’re on your way to becoming a subfi le programmer.

Oh Goody! Some Code

The complete source for this example at is available at http://www.mc-store.com/5104.
html, including compile information for a physical fi le (SFL001PF), a logical fi le 
(SFL001LF), a display fi le (SFL001DF), and the RPG (SFL001RG). To make this example 
work, you need to compile the three source members in order: fi rst SFL001PF, then 
SFL001DF, and fi nally SFL001RG. In real life, it might not always matter whether you 
compile the data fi le before the display fi le, since the data fi le will probably already 
exist. In this example, however, I reference the physical fi le fi eld information in my 
display fi le. As a result, the physical fi le needs to be created fi rst. Figure 1.2 shows the 
code for the physical fi le.

Figure 1.2: The DDS for the physical fi le SFL001PF.

After creating the physical fi le, you’ll need some sort of tool to get data into it. One 
way is to create a subfi le program to enter the data, but that’s what you’ll learn later in 
this book. For now, you can use the IBM i operating system’s (OS’s) native Data File 
Utility (DFU) or any third-party package your shop already owns. To start DFU, run the 
command UPDDTA FILE(SFL001PF), or take Option 18 next to SFL001PF if you’re using 
the Work with Objects Using Programming Development Manager (PDM) view.

The DDS for the physical fi le is pretty self-explanatory, so I will leave it alone. I’m 
going to start with the DDS for the display fi le. This is really where most of the work 
related to subfi les is accomplished. As I mentioned in the introduction, I’m assuming 
you have some basic knowledge of DDS. With that, let’s jump to the subfi le record 

 A                                      UNIQUE

 A          R PFR

 A            DBIDNM         7  0

 A            DBFNAM        20

 A            DBLNAM        20

 A            DBMINI         1

 A            DBNNAM        20

 A            DBADD1        30

 A            DBADD2        30

 A            DBADD3        30

 A          K DBIDNM

format, signifi ed by a record name of SFL1 and the record-level keyword of SFL. SFL1 
is where you defi ne the fi elds contained in each individual subfi le record. It is also 
the format that will be written to in your RPG program. This is the format that will 
actually contain the subfi le data.

The example in Figure 1.3 describes four fi elds: DBLNAM, DBFNAM, DBMINI, and 
DBNNAM. These fi elds are referenced from the physical fi le SFL001PF. It’s important to 
notice here that these fi elds are defi ned as output only (indicated by the “O” in position 
38), and they are set to start on row 5. That means the list of data to be displayed on 
the screen will start on row 5.

Each fi eld has its own starting column number. DBLNAM starts in column 2, DBFNAM 
starts in column 26, and so on. The column settings let you place the fi elds across the 
row in whatever fashion you want, just as you would with a printed report. That’s 
about it for the subfi le record format.

Figure 1.3: The DDS for the subfi le record format SFL1 in SFL001DF.

You have defi ned your fi eld’s size and usage, and you have designed the layout of each 
fi eld. The next record format is the subfi le control record format, signifi ed by the record 
name SF1CTL and the keyword SFLCTL. The subfi le control format must come directly 
after the subfi le record format. It’s always distinguished by the SFLCTL keyword. Even 
though it has to come directly after the SFL format, you still must place the name of the 
subfi le record format, in this case SFL1, within parentheses next to the SFLCTL keyword.

A                                      DSPSIZ(24 80 *DS3)

A                                      PRINT

A                                      ERRSFL

A                                      CA03

A                                      CA12

A*

A          R SFL1                      SFL

A*

A            DBLNAM    R        O  5 2REFFLD(PFR/DBLNAM*LIBL/SFL001PF)

A            DBFNAM    R        O  5 26REFFLD(PFR/DBFNAM *LIBL/SFL001PF)

A            DBMINI    R        O  5 50REFFLD(PFR/DBMINI *LIBL/SFL001PF)

A            DBNNAM    R        O  5 55REFFLD(PFR/DBNNAM *LIBL/SFL001PF)

The DDS for SF1CTL is shown in Figure 1.4. The subfi le record format is where you 
defi ne everything about the individual record, while the subfi le control record format 
is where you defi ne everything about the subfi le as a whole. You have two steps to take 
in the subfi le control record format:

1. Defi ne the subfi le characteristics with a series of keywords.
2. Defi ne any fi elds that will exist in the heading of the subfi le.

The fi elds usually include column headings for the subfi le records, as well as a title, 
program name, and date and time. However, they can also include input-capable fi elds, 
discussed in the next chapter. It’s important to note that fi elds defi ned in the subfi le 
control format are not part of the scrollable, subfi le records.

Figure 1.4: The DDS for the subfi le control record format SF1CTL in SFL001DF.

A          R SF1CTL                    SFLCTL(SFL1)

A*

A                                      SFLSIZ(0500)

A                                      SFLPAG(0017)

A                                      OVERLAY

A N32                                  SFLDSP

A N31                                  SFLDSPCTL

A  31                                  SFLCLR

A  90                                  SFLEND(*MORE)

A            RRN1           4S 0H      SFLRCDNBR

A                                  4  2'Last Name'

A                                      DSPATR(HI)

A                                  4 26'First Name'

A                                      DSPATR(HI)

A                                  4 50'MI'

A                                      DSPATR(HI)

A                                  4 55'Nick Name'

A                                      DSPATR(HI)

A                                  1  2'SFL001RG'

A                                  1 27'Simple Subfile Program'

A                                      DSPATR(HI)

A                                  1 71DATE

A                                      EDTCDE(Y)

A                                  2 71TIME

There is no requirement for the content of the fi elds in the subfi le control record 
format, but there is a limit as to where they can be placed. Remember that the SFL 
record format started on row 5. That means that fi elds defi ned in the SFLCTL record 
format must not be placed anywhere below row 4. Now, looking at the example, let’s 
examine the keywords used. They can be placed in any order within the SFLCTL record 
format, but they must also be entered before any fi elds or constants are defi ned.

The fi rst thing I do in Figure 1.4 is defi ne the subfi le size (SFLSIZ) and subfi le page size 
(SFLPAG). The SFLSIZ is set to 500. That means the initial number of records my subfi le 
can contain is 500. I can extend the number contained in the SFLSIZ keyword up to a 
maximum of 9,999 records, but I will cover that in the next chapter.

The SFLPAG keyword determines how many records can fi t on one page. I set mine to 
17, meaning that each screen of data will contain a maximum of 17 records. You’re 
limited on this parameter only by the number of records that can be physically 
displayed on a screen.

The OVERLAY keyword tells the display fi le to display the screen on top of what is 
already displayed—overlaying anything that is already there, but not erasing it. The 
subfi le display (SFLDSP) keyword displays the subfi le on the screen. It’s conditioned, in 
this case, by N32, which means it will display only when indicator 32 is set off. You’re 
not required to condition the SFLDSP keyword. I condition it in this example (and in 
most of my subfi le programs, for that matter) because I want to stop the subfi le from 
displaying if there are no records in it. You’ll get a runtime error if you try to display 
a subfi le with no records in it. There are a number of ways to handle the no-record 
situation, but I choose to set on indicator 32. This stops the subfi le, and subsequent 
errors, from being displayed.

The next keyword is the subfi le display control (SFLDSPCTL). This keyword allows 
you to display the control record format. The conditioning indicator on the SFLDSPCTL 
keyword, N31, conditions when the SFLCTL record format is to be displayed. In this 
case, the SFLCTL record format will be displayed only when it’s told to do so by the 
RPG program, and when indicator 31 is off. This keyword isn’t required if there are no 
fi elds to display or function keys to control in the format. Many keywords change the 

characteristics of the subfi le, but these changes aren’t seen until either the SFLDSP or 
the SFLDSPCTL keyword is activated.

The next keyword is the subfi le clear (SFLCLR), which clears the subfi le of its entries. 
It provides you with an empty subfi le just waiting to have records written to it. Notice 
that I used the opposite conditioning indicator for the SFLDSPCTL keyword. I did this 
because the same RPG operation (WRITE) is used to display the control record format 
and clear the subfi le. I probably don’t want to clear the subfi le at the same time I want 
to display it, so I use the same indicator. When indicator 31 is off, I want to display the 
subfi le. When it’s on, I want to clear it.

The subfi le end (SFLEND) keyword tells the user there are more records in the subfi le. A 
conditioning indicator is required when you’re using the SFLEND keyword. In this case, 
indicator 90 conditions the SFLEND keyword. The *MORE parameter is just one of the 
valid parameters used with the SFLEND keyword. (The others will be discussed later.) 
It will cause the screen to show the word “More . . .” at the bottom-right corner of the 
last subfi le record on the page. When the subfi le is on its last page, the word “Bottom” 
replaces “More . . .” to indicate that this is the end of the subfi le. SFLEND specifi ed 
by itself, with no parameter, will cause the screen to display a plus sign (“+”) in the 
lower-right corner when there are further pages to be displayed. When the subfi le is on 
its last page, the plus sign is replaced by a blank.

Now that all the keywords have been defi ned, you can defi ne any fi elds you need 
for your subfi le control record format. In Figure 1.4, I describe some basic column 
headings: a title, the date, and the time. Because this is the same technique you 
would use to defi ne headings on any display record format, it does not warrant 
discussion here.

Notice one particular fi eld defi ned in the SFLCTL record format: RRN1, which will be 
used as my subfi le relative record number. I defi ne it as a 4-digit signed numeric, with 
no decimal places. (Remember that there is a maximum of 9,999 records in a subfi le.) 
The “H” in position 38 indicates that this is a hidden fi eld. I’m not going to display this 
fi eld anywhere on the screen, but it’s extremely important because it will keep track of 
which subfi le record I’m working with.

The last record format defi ned in this DDS is the function key line. On line 23 of the 
display, I will list the function keys available to the user. This is the reason I use the 
OVERLAY keyword in the SFLCTL record format. I will fi rst write the FKEY1 format, 
which will display on line 23, and then OVERLAY the FKEY1 format, but not erase it.

I use a separate record format to display the function keys because I want my 
function keys to appear at the bottom of the screen, as is standard with most IBM i 
display screens. I cannot place the function key constants in the subfi le record format 
because I don’t want them repeated with each subfi le record. I’m also restricted from 
placing the function key constants in the subfi le control format because doing so would 
violate the rule of the subfi le control format not overlapping the subfi le record format. 
The only way I could get away with placing the function key constants in the subfi le 
control format is if I wanted them to display at the top of the screen. The FKEY1 format 
is shown in Figure 1.5.

Figure 1.5: The DDS for function key record format FKEY1 in SFL001DF.

There, you have just learned how to build a basic subfi le display fi le. This is more than 
a watered-down version appropriate only for learning. It is an example of a typical 
display fi le that you would actually use to display a list of data to a user. You’ve done 
the hard part, believe it or not. Now let’s look at the RPG code necessary to load and 
display this subfi le.

A          R FKEY1

A*

A                                 23  2'F3=Exit'

A                                      COLOR(BLU)

A                                 23 12'F12=Cancel'

A                                      COLOR(BLU)

Figure 1.6 shows the fi le specifi cations (F specs). Notice that there are two fi les 
described. SFL001DF is the display fi le you just learned about. I defi ne it as a combined 
(C in column 17), full procedural (F in column 18), externally described (E in column 
22) workstation (WORKSTN) fi le. I further defi ne SFL001DF on the next line of code 
by using the SFILE keyword with two parameters separated by a colon. The SFILE 
keyword defi nes a subfi le contained in the display fi le. If you want to use that subfi le 
in your program, you would code a SFILE line for every subfi le record format contained 
in your DDS. The fi rst parameter of the SFILE keyword is the name of the subfi le 
record format, and the second is the fi eld that will contain the subfi le relative record 
number.

Figure 1.6: RPG F specs for program SFL001RG.

Remember that I defi ned the relative record number fi eld (RRN1) as a hidden fi eld 
in the SFLCTL record format. The important thing to note is that the hidden fi eld 
defi ned in the subfi le record format is the one placed in the second parameter of the 
SFILE keyword. I will use SFL001LF, the logical fi le built over SFL001PF, to load the 
subfi le.

The body of the program consists only of a main routine and one subroutine. The 
main routine, shown in Figure 1.7, is pretty simple. I fi rst execute the subfi le build 
routine (Build_Subfile) and then code the DOU loop that will process the screen until 
either the F3 or F12 key has been pressed. I set indicator 90 on before displaying the 

FSfl001df  cf   e             Workstn

F                                     Sfile(Sfl1:Rrn1)

FSfl001lf  if   e           k Disk

screen, so that the end of the subfi le will be indicated properly. Indicator 90 is also 
the indicator I conditioned the SFLEND keyword on. Therefore, when the screen that 
contains the last record in the subfi le is displayed, the word “Bottom” will appear in 
the lower-right corner.

Figure 1.7: The mainline code for RPG program SFL001RG.

 /Free

  // Main Routine

  Exsr Build_Subfile;   // Execute the subfile build routine.

  Dou  *Inkc or *Inkl;  // Process until F3 or F12 is pressed.

    *In90 = *On;        // Set Subfile end indicator.

    Write Fkey1;        // Display the function key line.

    Exfmt Sf1ctl;       // Display the subfile.

  Enddo;

  *Inlr = *On;

Let’s examine the Build_Subfile routine fi rst, since that code will be executed before the 
DOU loop. The subroutine Build_Subfile is shown in Figure 1.8.

 // ******************************************************************

 // Build_Subfile - Build the List

 // ******************************************************************

Begsr Build_Subfile;

 // Clear the Subfile

 Rrn1 = *Zero;       // Set the record counter to 0.

 *In31 = *On;        // Turn on the subfile clear indicator.

 Write Sf1ctl;       // Clear the subfile.

 *In31 = *Off;       // Turn off the subfile clear indicator.

 // Load data to subfile

 Setll (*Loval) Sfl001lf;

 Read Sfl001lf;

Continued

Figure 1.8: The subroutine used to build the subfi le in SFL001RG.

 Dow (Not %eof) And (Rrn1 <= 500); // Read while there are records in

                                   // the file, up to the 500 limit.

   Rrn1 = Rrn1 + 1;  // Increment subfile record number.

   Write Sfl1;       // Write the record to the subfile.

   Read Sfl001lf;    // Read next data record.

 Enddo;

 // If no records were loaded, do not display the subfile

 If Rrn1 = *Zero;    // If no records were loaded,

   *In32 = *On;      // set the indicator to NOT display the subfile.

 Else;               // Otherwise,

   Rrn1 = 1;         // set the rec to 1 so the first page is displayed.

 Endif;

 Endsr;

/End-Free

The fi rst block of code clears the subfi le (SFLCLR in the DDS). Because it was 
conditioned on indicator 31, I set on indicator 31 and write to the SFLCTL record 
format, which performs the clear. (The SFLDSPCTL isn’t activated in this case because 
it’s conditioned on indicator 31 being off.) After I write the SFLCTL record format, I set 
indicator 31 off, so the next write to the format will cause it to be displayed.

Now it’s time to load data to the subfi le. I do this by setting the fi le pointer to the 
beginning of the fi le and reading the fi rst record of the data fi le. The three lines 
contained within the DOW loop are suffi cient to load my subfi le. I’m going to process 
the DOW loop until there are no more records in the fi le, or until I hit the 500-record 
limit set on the SFLSIZ keyword. Each time I read a record, I increment the subfi le 
relative record number (RRN1) and write the record to the subfi le record format. 
Because my subfi le fi eld names are the same as the data fi le fi eld names, no MOVE, 
MOVEL, EVAL, or Z-ADD statements are necessary.

The IF statement below the loop keeps an error from ending the program. If no records 
were read from the fi le (RRN1 = 0), I set indicator 32 on. Remember that the subfi le 
display (SFLDSP) keyword is conditioned on 32 being off. If it’s on, the subfi le won’t 
be displayed. Displaying a subfi le with zero records will cause an error and end the 
program.

Back in the main routine (Figure 1.7), now that the subfi le is loaded, I will display it. 
Within the DOU loop, I write the function key format (FKEY1). This will display the 
function keys available to the user on line 23 of the screen. I then display the subfi le 
with the execute format (EXFMT) keyword.

When displaying a subfi le, the EXFMT is always done on the SFLCTL record format. 
Because the OS knows which subfi le record format is associated with which subfi le 
control record format, there’s no need to explicitly perform an EXFMT to the SFL 
record format. I set indicator 31 off after the subfi le was cleared, and with this write 



Chapter 1 The Specifi cs on Subfi les • 23

to the SFLCTL record format, the subfi le control record will be displayed. If there are 
records in your subfi le, indicator 32 will be off. As a result, the subfi le should also be 
displayed.

Depending on how many records you added to your data fi le, you would see 
either “More . . .” or “Bottom” in the lower-right corner of the screen. If you saw 
“More . . . ,” you’d be able to press the Page Down key to display the next page of 
data. You could do this until you saw “Bottom.” If there were more than one page of 
data, you’d be able to use the Page Up key to scroll back to the top.

What’s nice about this is that I didn’t have to code a single line of RPG to handle the 
scrolling. IBM i took care of it for me. The program will sit on the EXFMT operation 
until a key is pressed that returns control back to the RPG program. In this case, the 
Page Up and Page Down keys are handled by the system and won’t return control back 
to the program.

This method of subfi le programming is called the load-all method. Using this method, 
you load the subfi le one time, and display it to the screen. IBM i handles the paging 
for you. As you can see, it doesn’t take a lot of code to get a load-all subfi le up and 
running. In the next chapter, you’ll learn more about the load-all method as well as two 
other methods, and when to use which one.

Hip Hop Subfi les Can Wrap

There might be a time when one line of data doesn’t cut it. The subfi le in Figure 1.1 
displays fi rst name, last name, middle initial, and nickname neatly on one line. But 
what if you want to show part of the address on the screen? There isn’t enough room 
on the subfi le line to squeeze in the address. What do you do? Do you abandon the 
subfi le approach and try something else? The answer, thankfully, is no. With just a 
little more code in the DDS and, optionally, a couple more lines in the RPG, your 

subfi le can display multiple lines of data per subfi le record. Figure 1.9 shows the 
output from previous subfi le programs, with the addition of the address, city, and state 
on a second line.

Figure 1.9: An example of a subfi le with lines that wrap.

Just Add the Fields

Most of the work needed to create the multiple-line subfi le record is accomplished in 
the DDS. In fact, the only reason you might decide to modify the RPG is to tighten up 
the technique a little. You could easily leave the RPG alone and still provide multiple-
line subfi le records.

Figure 1.10 shows a new version of the subfi le record format that places two address 
fi elds on a second line. All I’ve done is add two new fi elds on a second line of the 
subfi le. The subfi le starts on line 5 with the name fi elds, and now extends to a second 
line (line 6) with the addition of DBADD1 and DBADD2. This will cause your subfi le to 
display two lines per record. The system fi gures all this out for you, once you defi ne 
your subfi le record format and determine how many records you want to display 

on a page (SFLPAG) in your record control format. You can see the complete DDS 
(SFL010DF) at http://www.mc-store.com/5104.html.

Figure 1.10: The subfi le record format for a multiple-line subfi le record.

This brings me to my next point. There is one additional step you’ll have to take to 
properly fi t this new subfi le on the screen. You need to change the SFLPAG keyword in 
the control record format. In the previous example, SFLPAG was set to 17. However, 
trying to fi t 17 double-lined subfi le records on a page will create some problems when 
you try to compile. To counter this, you’ll have to modify the SFLPAG number to some 
appropriate number. Because I had 17 before and am now displaying two lines per 
record, I changed the SFLPAG keyword to 8. That means the subfi le will display eight 
two-line subfi le records per page, which (and I know you can do the math) equates to 
16 actual lines.

Sfl fold and Sfl drop

By simply adding the two fi elds to the subfi le record format, changing the SFLPAG 
keyword to 8, and recompiling both the display fi le and the RPG program, your 
subfi le will now look something like Figure 1.9. Cool, right? Well, what if you want 
to toggle back and forth between one- and two-line subfi le records? Maybe you don’t 
always want to see the address information. Maybe you’d like to display the subfi le 
as in Figure 1.1 and press a function key to show the address lines (Figure 1.9 shows 
that hint).

Is this asking too much? Of course not. With the addition of a few subfi le keywords, 
you can actually toggle between single-line and multiple-line subfi le records.

A            DBLNAM    R        O  5 2REFFLD(PFR/DBLNAM *LIBL/SFL001PF)

A            DBFNAM    R        O  5 26REFFLD(PFR/DBFNAM *LIBL/SFL001PF)

A            DBMINI    R        O  5 50REFFLD(PFR/DBMINI *LIBL/SFL001PF)

A            DBNNAM    R        O  5 55REFFLD(PFR/DBNNAM *LIBL/SFL001PF)

A            DBADD1    R        O  6 5REFFLD(PFR/DBADD1 *LIBL/SFL001PF)

A            DBADD2    R        O  6 37REFFLD(PFR/DBADD2 *LIBL/SFL001PF)


